In [1]:
```
import Pkg
Pkg.activate(@__DIR__)
Pkg.instantiate()
import FiniteDiff
import ForwardDiff as FD
import Convex as cvx
import ECOS
using LinearAlgebra
using Plots
using Random
using JLD2
using Test
using MeshCat
const mc = MeshCat
using TrajOptPlots
using StaticArrays
using Printf
```

```
  Activating environment at `~/ocrl_ws/16745-ocrl/HW4_S23-new/Project.toml`
```

In [2]:
```
include(joinpath(@__DIR__, "utils","ilc_visualizer.jl"))
```

Out[2]:
```
vis_traj! (generic function with 1 method)
```

Q1: Iterative Learning Control (ILC) (40 pts)

In this problem, you will use ILC to generate a control trajectory for a Car as it swerves to avoid a moose, also known as "the moose test" (wikipedia, video). We will model the dynamics of the car as with a simple nonlinear bicycle model, with the following state and control:

$$x = \begin{bmatrix} p_x \\ p_y \\ \theta \\ \delta \\ v \\ \omega \end{bmatrix}, \qquad u = \begin{bmatrix} a \\ \dot{\delta} \end{bmatrix} \tag{1}$$

where p_x and p_y describe the 2d position of the bike, θ is the orientation, δ is the steering angle, and v is the velocity. The controls for the bike are acceleration a, and steering angle rate $\dot{\delta}$.

In [3]:
```
function estimated_car_dynamics(model::NamedTuple, x::Vector, u::Vector)::Vector
    # nonlinear bicycle model continuous time dynamics
    px, py, θ, δ, v = x
    a, δdot = u

    β = atan(model.lr * δ, model.L)
    s,c = sincos(θ + β)
    ω = v*cos(β)*tan(δ) / model.L

    vx = v*c
    vy = v*s

    xdot = [
        vx,
        vy,
        ω,
        δdot,
        a
    ]

    return xdot
end
function rk4(model::NamedTuple, ode::Function, x::Vector, u::Vector, dt::Real)::Vector
    k1 = dt * ode(model, x,        u)
    k2 = dt * ode(model, x + k1/2, u)
    k3 = dt * ode(model, x + k2/2, u)
    k4 = dt * ode(model, x + k3,   u)
    return x + (1/6)*(k1 + 2*k2 + 2*k3 + k4)
end
```

Out[3]:
```
rk4 (generic function with 1 method)
```

We have computed an optimal trajectory X_{ref} and U_{ref} for a moose test trajectory offline using this `estimated_car_dynamics` function. Unfortunately, this is a highly approximate dynamics model, and when we run U_{ref} on the car, we get a very different trajectory than we expect. This is caused by a significant sim to real gap. Here we will show what happens when we run these controls on the true dynamics:

In [4]:
```
function load_car_trajectory()
    # load in trajectory we computed offline
    path = joinpath(@__DIR__, "utils","init_control_car_ilc.jld2")
    F = jldopen(path)
```

```
    Xref = F["X"]
    Uref = F["U"]
    close(F)
    return Xref, Uref
end
function true_car_dynamics(model::NamedTuple, x::Vector, u::Vector)::Vector
    # true car dynamics
    px, py, θ, δ, v = x
    a, δdot = u

    # sluggish controls (not in the approximate version)
    a = 0.9*a - 0.1
    δdot = 0.9*δdot - .1*δ + .1

    β = atan(model.lr * δ, model.L)
    s,c = sincos(θ + β)
    ω = v*cos(β)*tan(δ) / model.L

    vx = v*c
    vy = v*s

    xdot = [
        vx,
        vy,
        ω,
        δdot,
        a
    ]

    return xdot
end

@testset "sim to real gap" begin
    # problem size
    nx = 5
    nu = 2
    dt = 0.1
    tf = 5.0
    t_vec = 0:dt:tf
    N = length(t_vec)
    model = (L = 2.8, lr = 1.6)

    # optimal trajectory computed offline with approximate model
    Xref, Uref = load_car_trajectory()

    # TODO: simulated Uref with the true car dynamics and store the states in Xsim
    Xsim = [zeros(nx) for i = 1:N]

    Xsim[1] = Xref[1]

    for k = 1:(N-1)
        Xsim[k+1] = rk4(model, true_car_dynamics, Xsim[k], Uref[k], dt)
    end

    # --------testing------------------
    @test norm(Xsim[1] - Xref[1]) == 0
    @test norm(Xsim[end] - [3.26801052, 15.0590156, 2.0482790, 0.39056168, 4.5],Inf) < 1e-4

    # --------plotting/animation-------
    Xm= hcat(Xsim...)
    Xrefm = hcat(Xref...)
    plot(Xrefm[1,:], Xrefm[2,:], ls = :dash, label = "reference",
         xlabel = "x (m)", ylabel = "y (m)", title = "Simulation vs Reference")
    display(plot!(Xm[1,:], Xm[2,:], label = "actual"))

end
```




```
Test Summary:   | Pass  Total
sim to real gap |    2      2
```

Out[4]: `Test.DefaultTestSet("sim to real gap", Any[], 2, false, false)`

In order to account for this, we are going to use ILC to iteratively correct our control until we converge.

To encourage the trajectory of the bike to follow the reference, the objective value for this problem is the following:

$$J(X,U) = \sum_{i=1}^{N-1} \left[\frac{1}{2}(x_i - x_{ref,i})^T Q (x_i - x_{ref,i}) + \frac{1}{2}(u_i - u_{ref,i})^T R (u_i - u_{ref,i}) \right] + \frac{1}{2}(x_N - x_{ref,N})^T Q_f (x_N - x_{ref,N})$$

Using ILC as described in Lecture 18, we are to linearize our approximate dynamics model about X_{ref} and U_{ref} to get the following Jacobians:

$$A_k = \frac{\partial f}{\partial x}\bigg|_{x_{ref,k}, u_{ref,k}}, \qquad B_k = \frac{\partial f}{\partial u}\bigg|_{x_{ref,k}, u_{ref,k}}$$

where $f(x,u)$ is our **approximate discrete** dynamics model (`estimated_car_dynamics` + `rk4`). **You will form these Jacobians exactly once, using `Xref` and `Uref`.** Here is a summary of the notation:

- X_{ref} (`Xref`) - Optimal trajectory computed offline with approximate dynamics model.
- U_{ref} (`Uref`) - Optimal controls computed offline with approximate dynamics model.
- X_{sim} (`Xsim`) - Simulated trajectory with real dynamics model.
- $\bar{U}$ (`Ubar`) - Control we use for simulation with real dynamics model (this is what ILC updates).

In the second step of ILC, we solve the following optimization problem:

$$\begin{aligned} \min_{\Delta x_{1:N}, \Delta u_{1:N-1}} \quad & J(X_{sim} + \Delta X, \bar{U} + \Delta U) && (2) \\ \text{st} \quad & \Delta x_1 = 0 && (3) \\ & \Delta x_{k+1} = A_k \Delta x_k + B_k \Delta u_k \quad \text{for } k = 1, 2, \ldots, N-1 && (4) \end{aligned}$$

We are going to initialize our $\bar{U}$ with U_{ref}, then the ILC algorithm will update $\bar{U} = \bar{U} + \Delta U$ at each iteration. It should only take 5-10 iterations to converge down to $\|\Delta U\| < 1 \cdot 10^{-2}$. You do not need to do any sort of linesearch between ILC updates.

In [5]:
```
# feel free to use/not use any of these

function trajectory_cost(Xsim::Vector{Vector{Float64}}, # simulated states
                         Ubar::Vector{Vector{Float64}}, # simulated controls (ILC iterates this)
                         Xref::Vector{Vector{Float64}}, # reference X's we want to track
                         Uref::Vector{Vector{Float64}}, # reference U's we want to track
                         Q::Matrix,                     # LQR tracking cost term
                         R::Matrix,                     # LQR tracking cost term
                         Qf::Matrix                     # LQR tracking cost term
                         )::Float64                     # return cost J

    J = 0
    # TODO: return trajectory cost J(Xsim, Ubar)
    N = length(Xsim)
    for i = 1:(N-1)
        J += 0.5 * (Xsim[i] - Xref[i])' * Q * (Xsim[i] - Xref[i])
        J += 0.5 * (Ubar[i] - Uref[i])' * R * (Ubar[i] - Uref[i])
    end
    J += 0.5 * (Xsim[N] - Xref[N])' * Qf * (Xsim[N] - Xref[N])
    return J
end

function vec_from_mat(Xm::Matrix)::Vector{Vector{Float64}}
```

```julia
    # convert a matrix into a vector of vectors
    X = [Xm[:,i] for i = 1:size(Xm,2)]
    return X
end

function ilc_update(Xsim::Vector{Vector{Float64}}, # simulated states
                    Ubar::Vector{Vector{Float64}}, # simulated controls (ILC iterates this)
                    Xref::Vector{Vector{Float64}}, # reference X's we want to track
                    Uref::Vector{Vector{Float64}}, # reference U's we want to track
                    As::Vector{Matrix{Float64}},   # vector of A jacobians at each time step
                    Bs::Vector{Matrix{Float64}},   # vector of B jacobians at each time step
                    Q::Matrix,                     # LQR tracking cost term
                    R::Matrix,                     # LQR tracking cost term
                    Qf::Matrix                     # LQR tracking cost term
                    )::Vector{Vector{Float64}}     # return vector of ΔU's

    # solve optimization problem for ILC update
    N = length(Xsim)
    nx,nu = size(Bs[1])

    # create variables
    ΔX = cvx.Variable(nx, N)
    ΔU = cvx.Variable(nu, N-1)

    # TODO: cost function (tracking cost on Xref, Uref)
#     cost = trajectory_cost(Xsim, Ubar, Xref, Uref, Q, R ,Qf)

    cost = 0

    for k = 1:(N-1)
        xk = Xsim[k] + ΔX[:,k] - Xref[k]
        uk = Ubar[k] + ΔU[:,k] - Uref[k]
        cost += (0.5 * cvx.quadform(xk,Q))
        cost += (0.5 * cvx.quadform(uk,R))
    end

    # add terminal cost
    xn = Xsim[N] + ΔX[:,N] - Xref[N]
    cost += (0.5 * cvx.quadform(xn,Qf))

    # problem instance
    prob = cvx.minimize(cost)

    # TODO: initial condition constraint
    prob.constraints += (ΔX[:,1] == zeros(nx))

    # TODO: dynamics constraints
    for k = 1:(N-1)
        # dynamics constraints
        prob.constraints += (ΔX[:,k+1] == (As[k] * ΔX[:,k] + Bs[k]*ΔU[:,k]))
    end

    cvx.solve!(prob, ECOS.Optimizer; silent_solver = true)

    # return ΔU
    ΔU = vec_from_mat(ΔU.value)

    return ΔU
end
```

Out[5]:

```
ilc_update (generic function with 1 method)
```

Here you will run your ILC algorithm. The resulting plots should show the simulated trajectory `Xsim` tracks `Xref` very closely, but there should be a significant difference between `Uref` and `Ubar`.

In [6]:

```julia
@testset "ILC" begin

    # problem size
    nx = 5
    nu = 2
    dt = 0.1
    tf = 5.0
    t_vec = 0:dt:tf
    N = length(t_vec)

    # optimal trajectory computed offline with approximate model
    Xref, Uref = load_car_trajectory()

    # initial and terminal conditions
    xic = Xref[1]
    xg = Xref[N]

    # LQR tracking cost to be used in ILC
    Q = diagm([1,1,.1,.1,.1])
    R = .1*diagm(ones(nu))
    Qf = 1*diagm(ones(nx))
```

```
    # load all useful things into params
    model = (L = 2.8, lr = 1.6)

    params = (Q = Q, R = R, Qf = Qf,xic = xic, xg = xg, Xref=Xref,Uref=Uref,
          dt = dt,
          N = N,
          model = model)


    # this holds the sim trajectory (with real dynamics)
    Xsim = [zeros(nx) for i = 1:N]

    # this is the feedforward control ILC is updating
    Ubar = [zeros(nu) for i = 1:(N-1)]
    Ubar .= Uref # initialize Ubar with Uref

    # TODO: calculate Jacobians
    A = [zeros(nx,nx) for i = 1:(N-1)]
    B = [zeros(nx,nu) for i = 1:(N-1)]
    for k = 1:(N-1)
        A[k] = FD.jacobian(x->rk4(model, estimated_car_dynamics, x, Uref[k], dt),Xref[k])
        B[k] = FD.jacobian(u->rk4(model, estimated_car_dynamics, Xref[k], u, dt),Uref[k])
    end


    # logging stuff
    @printf "iter     objv        |ΔU|     \n"
    @printf "-------------------------------\n"

    for ilc_iter = 1:10 # it should not take more than 10 iterations to converge
        # TODO: rollout
        Xsim[1] = Xref[1]
        for k = 1:(N-1)
            U = Ubar[k]
            Xsim[k+1] = rk4(model, true_car_dynamics, Xsim[k], U, dt)
        end

        # TODO: calculate objective val (trajectory_cost)
        obj_val = trajectory_cost(Xsim, Ubar, Xref, Uref, Q, R ,Qf)

        # solve optimization problem for update (ilc_update)
        ΔU = ilc_update(Xsim, Ubar, Xref, Uref, A, B, Q, R ,Qf)

        # TODO: update the control
        for k = 1 : (N-1)
            Ubar[k] = Ubar[k] + ΔU[k]
        end

        # logging
        @printf("%3d   %10.3e  %10.3e  \n", ilc_iter, obj_val, sum(norm.(ΔU)))

    end

    # --------plotting/animation-------
    Xm= hcat(Xsim...)
    Um = hcat(Ubar...)
    Xrefm = hcat(Xref...)
    Urefm = hcat(Uref...)
    plot(Xrefm[1,:], Xrefm[2,:], ls = :dash, label = "reference",
         xlabel = "x (m)", ylabel = "y (m)", title = "Trajectory")
    display(plot!(Xm[1,:], Xm[2,:], label = "actual"))

    plot(t_vec[1:end-1], Urefm', ls = :dash, lc = [:green :blue],label = "",
         xlabel = "time (s)", ylabel = "controls", title = "Controls (-- is reference)")
    display(plot!(t_vec[1:end-1], Um', label = ["δ̇" "a"], lc = [:green :blue]))

    # animation
    vis = Visualizer()
    X_vis = [[x[1],x[2],0.1] for x in Xsim]
    vis_traj!(vis, :traj, X_vis; R = 0.02)
    vis_model = TrajOptPlots.RobotZoo.BicycleModel()
    TrajOptPlots.set_mesh!(vis, vis_model)
    X = [x[SA[1,2,3,4]] for x in Xsim]
    visualize!(vis, vis_model, tf, X)
    display(render(vis))

    # ----------testing-------------
    @test 0.1 <= sum(norm.(Xsim - Xref)) <= 1.0 # should be ~0.7
    @test 5 <=  sum(norm.(Ubar - Uref)) <= 10 # should be ~7.7

end
```

```
iter     objv        |ΔU|
------------------------------
  1    1.436e+03   6.307e+01
  2    1.131e+03   4.498e+01
  3    5.122e+02   9.266e+01
  4    6.109e+00   1.394e+01
  5    4.614e-01   1.959e+00
  6    7.746e-02   1.679e-01
  7    7.157e-02   1.649e-02
  8    7.144e-02   1.578e-03
  9    7.143e-02   1.911e-04
 10    7.143e-02   3.029e-05
```






```
┌ Info: MeshCat server started. You can open the visualizer by visiting the following URL in your browser:
└ http://127.0.0.1:8706
```



Open Controls

```
Test Summary: | Pass  Total
ILC           |    2      2
```

Out[6]:
```
Test.DefaultTestSet("ILC", Any[], 2, false, false)
```

In [ ]:

In [ ]:

In [1]:
```
import Pkg
Pkg.activate(@__DIR__)
Pkg.instantiate()
import MathOptInterface as MOI
import Ipopt
import FiniteDiff
import ForwardDiff as FD
import Convex as cvx
import ECOS
using LinearAlgebra
using Plots
using Random
using JLD2
using Test
using MeshCat
const mc = MeshCat
using TrajOptPlots
using StaticArrays
using Printf
```

```
  Activating environment at `~/ocrl_ws/16745-ocrl/HW4_S23-new/Project.toml`
```

In [2]:
```
include(joinpath(@__DIR__, "utils","fmincon.jl"))
include(joinpath(@__DIR__, "utils","walker.jl"))
```

Out[2]:
```
update_walker_pose! (generic function with 1 method)
```

(If nothing loads here, check out `walker.gif` in the repo)

NOTE: This question will have long outputs for each cell, remember you can use `cell -> all output -> toggle scrolling` to better see it all

Q2: Hybrid Trajectory Optimization (60 pts)

In this problem you'll use a direct method to optimize a walking trajectory for a simple biped model, using the hybrid dynamics formulation. You'll pre-specify a gait sequence and solve the problem using Ipopt. Your final solution should look like the video above.

The Dynamics

Our system is modeled as three point masses: one for the body and one for each foot. The state is defined as the x and y positions and velocities of these masses, for a total of 6 degrees of freedom and 12 states. We will label the position and velocity of each body with the following notation:

$$r^{(b)} = \begin{bmatrix} p_x^{(b)} \\ p_y^{(b)} \end{bmatrix} \qquad v^{(b)} = \begin{bmatrix} v_x^{(b)} \\ v_y^{(b)} \end{bmatrix} \tag{1}$$

$$r^{(1)} = \begin{bmatrix} p_x^{(1)} \\ p_y^{(1)} \end{bmatrix} \qquad v^{(1)} = \begin{bmatrix} v_x^{(1)} \\ v_y^{(1)} \end{bmatrix} \tag{2}$$

$$r^{(2)} = \begin{bmatrix} p_x^{(2)} \\ p_y^{(2)} \end{bmatrix} \qquad v^{(2)} = \begin{bmatrix} v_x^{(2)} \\ v_y^{(2)} \end{bmatrix} \tag{3}$$

Each leg is connected to the body with prismatic joints. The system has three control inputs: a force along each leg, and the torque between the legs.

The state and control vectors are ordered as follows:

$$x = \begin{bmatrix} p_x^{(b)} \\ p_y^{(b)} \\ p_x^{(1)} \\ p_y^{(1)} \\ p_x^{(2)} \\ p_y^{(2)} \\ v_x^{(b)} \\ v_y^{(b)} \\ v_x^{(1)} \\ v_y^{(1)} \\ v_x^{(2)} \\ v_y^{(2)} \end{bmatrix} \qquad u = \begin{bmatrix} F^{(1)} \\ F^{(2)} \\ \tau \end{bmatrix}$$

where e.g. $p_x^{(b)}$ is the x position of the body, $v_y^{(i)}$ is the y velocity of foot i, $F^{(i)}$ is the force along leg i, and τ is the torque between the legs.

The continuous time dynamics and jump maps for the two stances are shown below:

In [3]:
```julia
function stance1_dynamics(model::NamedTuple, x::Vector, u::Vector)
    # dynamics when foot 1 is in contact with the ground

    mb,mf = model.mb, model.mf
    g = model.g

    M = Diagonal([mb mb mf mf mf mf])

    rb  = x[1:2]   # position of the body
    rf1 = x[3:4]   # position of foot 1
    rf2 = x[5:6]   # position of foot 2
    v   = x[7:12]  # velocities


    ℓ1x = (rb[1]-rf1[1])/norm(rb-rf1)
    ℓ1y = (rb[2]-rf1[2])/norm(rb-rf1)
    ℓ2x = (rb[1]-rf2[1])/norm(rb-rf2)
    ℓ2y = (rb[2]-rf2[2])/norm(rb-rf2)

    B = [ℓ1x  ℓ2x  ℓ1y-ℓ2y;
         ℓ1y  ℓ2y  ℓ2x-ℓ1x;
          0    0     0;
          0    0     0;
          0  -ℓ2x  ℓ2y;
          0  -ℓ2y -ℓ2x]

    v̇ = [0; -g; 0; 0; 0; -g] + M\(B*u)

    ẋ = [v; v̇]

    return ẋ
end

function stance2_dynamics(model::NamedTuple, x::Vector, u::Vector)
    # dynamics when foot 2 is in contact with the ground

    mb,mf = model.mb, model.mf
    g = model.g
    M = Diagonal([mb mb mf mf mf mf])

    rb  = x[1:2]   # position of the body
    rf1 = x[3:4]   # position of foot 1
    rf2 = x[5:6]   # position of foot 2
    v   = x[7:12]  # velocities

    ℓ1x = (rb[1]-rf1[1])/norm(rb-rf1)
    ℓ1y = (rb[2]-rf1[2])/norm(rb-rf1)
    ℓ2x = (rb[1]-rf2[1])/norm(rb-rf2)
    ℓ2y = (rb[2]-rf2[2])/norm(rb-rf2)

    B = [ℓ1x  ℓ2x  ℓ1y-ℓ2y;
         ℓ1y  ℓ2y  ℓ2x-ℓ1x;
        -ℓ1x   0  -ℓ1y;
        -ℓ1y   0   ℓ1x;
          0    0    0;
          0    0    0]

    v̇ = [0; -g; 0; -g; 0; 0] + M\(B*u)

    ẋ = [v; v̇]

    return ẋ
end

function jump1_map(x)
    # foot 1 experiences inelastic collision
    xn = [x[1:8]; 0.0; 0.0; x[11:12]]
    return xn
end

function jump2_map(x)
    # foot 2 experiences inelastic collision
    xn = [x[1:10]; 0.0; 0.0]
    return xn
end

function rk4(model::NamedTuple, ode::Function, x::Vector, u::Vector, dt::Real)::Vector
    k1 = dt * ode(model, x,        u)
    k2 = dt * ode(model, x + k1/2, u)
    k3 = dt * ode(model, x + k2/2, u)
    k4 = dt * ode(model, x + k3,   u)
```

```
    return x + (1/6)*(k1 + 2*k2 + 2*k3 + k4)
end
```

Out[3]: rk4 (generic function with 1 method)

We are setting up this problem by scheduling out the contact sequence. To do this, we will define the following sets:

$$\mathcal{M}_1 = \{1{:}5, 11{:}15, 21{:}25, 31{:}35, 41{:}45\} \tag{4}$$
$$\mathcal{M}_2 = \{6{:}10, 16{:}20, 26{:}30, 36{:}40\} \tag{5}$$

where $\mathcal{M}_1$ contains the time steps when foot 1 is pinned to the ground (`stance1_dynamics`), and $\mathcal{M}_2$ contains the time steps when foot 2 is pinned to the ground (`stance2_dynamics`). The jump map sets $\mathcal{J}_1$ and $\mathcal{J}_2$ are the indices where the mode of the next time step is different than the current, i.e. $\mathcal{J}_i \equiv \{k+1 \notin \mathcal{M}_i \mid k \in \mathcal{M}_i\}$. We can write these out explicitly as the following:

$$\mathcal{J}_1 = \{5, 15, 25, 35\} \tag{6}$$
$$\mathcal{J}_2 = \{10, 20, 30, 40\} \tag{7}$$

Another term you will see is set subtraction, or $\mathcal{M}_i \setminus \mathcal{J}_i$. This just means that if $k \in \mathcal{M}_i \setminus \mathcal{J}_i$, then k is in $\mathcal{M}_i$ but not in $\mathcal{J}_i$.

We will make use of the following Julia code for determining which set an index belongs to:

In [4]:
```
let
    M1 = vcat([ (i-1)*10      .+ (1:5)   for i = 1:5]...) # stack the set into a vector
    M2 = vcat([((i-1)*10 + 5) .+ (1:5)   for i = 1:4]...) # stack the set into a vector
    J1 = [5,15,25,35]
    J2 = [10,20,30,40]

    @show M1
    @show M2

    @show  (5 in M1) # show if 5 is in M1
    @show  (5 in J1) # show if 5 is in J1
    @show !(5 in M1) # show is 5 is not in M1

    @show (5 in M1) && !(5 in J1) # 5 in M1 but not J1 (5 ∈ M_1 \ J1)

end
```

```
M1 = [1, 2, 3, 4, 5, 11, 12, 13, 14, 15, 21, 22, 23, 24, 25, 31, 32, 33, 34, 35, 41, 42, 43, 44, 45]
M2 = [6, 7, 8, 9, 10, 16, 17, 18, 19, 20, 26, 27, 28, 29, 30, 36, 37, 38, 39, 40]
5 in M1 = true
5 in J1 = true
!(5 in M1) = false
5 in M1 && !(5 in J1) = false
```

Out[4]: false

We are now going to setup and solve a constrained nonlinear program. The optimization problem looks complicated but each piece should make sense and be relatively straightforward to implement. First we have the following LQR cost function that will track x_{ref} (`Xref`) and u_{ref} (`Uref`):

$$J(x_{1:N}, u_{1:N-1}) = \sum_{i=1}^{N-1} \left[\frac{1}{2}(x_i - x_{ref,i})^T Q (x_i - x_{ref,i}) + \frac{1}{2}(u_i - u_{ref,i})^T R (u_i - u_{ref,i})\right] + \frac{1}{2}(x_N - x_{ref,N})^T Q_f (x_N - x_{ref,N})$$

Which goes into the following full optimization problem:

$$\begin{aligned}
\min_{x_{1:N}, u_{1:N-1}} \quad & J(x_{1:N}, u_{1:N-1}) && && (8)\\
\text{st} \quad & x_1 = x_{ic} && && (1)\\
& x_N = x_g && && (2)\\
& x_{k+1} = f_1(x_k, u_k) && \text{for } k \in \mathcal{M}_1 \setminus \mathcal{J}_1 && (3)\\
& x_{k+1} = f_2(x_k, u_k) && \text{for } k \in \mathcal{M}_2 \setminus \mathcal{J}_2 && (4)\\
& x_{k+1} = g_2(f_1(x_k, u_k)) && \text{for } k \in \mathcal{J}_1 && (5)\\
& x_{k+1} = g_1(f_2(x_k, u_k)) && \text{for } k \in \mathcal{J}_2 && (6)\\
& x_k[4] = 0 && \text{for } k \in \mathcal{M}_1 && (7)\\
& x_k[6] = 0 && \text{for } k \in \mathcal{M}_2 && (8)\\
& 0.5 \leq \|r_k^{(b)} - r_k^{(1)}\|_2 \leq 1.5 && \text{for } k \in [1, N] && (9)\\
& 0.5 \leq \|r_k^{(b)} - r_k^{(2)}\|_2 \leq 1.5 && \text{for } k \in [1, N] && (10)\\
& x_k[2, 4, 6] \geq 0 && \text{for } k \in [1, N] && (11)
\end{aligned}$$

Each constraint is now described, with the type of constraint for `fmincon` in parantheses:

1. Initial condition constraint **(equality constraint)**.
2. Terminal condition constraint **(equality constraint)**.
3. Stance 1 discrete dynamics **(equality constraint)**.
4. Stance 2 discrete dynamics **(equality constraint)**.
5. Discrete dynamics from stance 1 to stance 2 with jump 2 map **(equality constraint)**.
6. Discrete dynamics from stance 2 to stance 1 with jump 1 map **(equality constraint)**.

7. Make sure the foot 1 is pinned to the ground in stance 1 **(equality constraint)**.
8. Make sure the foot 2 is pinned to the ground in stance 2 **(equality constraint)**.
9. Length constraints between main body and foot 1 **(inequality constraint)**.
10. Length constraints between main body and foot 2 **(inequality constraint)**.
11. Keep the y position of all 3 bodies above ground **(primal bound)**.

And here we have the list of mathematical functions to the Julia function names:

- f_1 is `stance1_dynamics` + `rk4`
- f_2 is `stance2_dynamics` + `rk4`
- g_1 is `jump1_map`
- g_2 is `jump2_map`

For instance, $g_2(f_1(x_k, u_k))$ is `jump2_map(rk4(model, stance1_dynamics, xk, uk, dt))`

Remember that $r^{(b)}$ is defined above.

In [5]:
```
function reference_trajectory(model, xic, xg, dt, N)
    # creates a reference Xref and Uref for walker

    Uref = [[model.mb*model.g*0.5;model.mb*model.g*0.5;0] for i = 1:(N-1)]

    Xref = [zeros(12) for i = 1:N]

    horiz_v = (3/N)/dt
    xs = range(-1.5, 1.5, length = N)
    Xref[1] = 1*xic
    Xref[N] = 1*xg

    for i = 2:(N-1)
        Xref[i] = [xs[i],1,xs[i],0,xs[i],0,horiz_v,0,horiz_v,0,horiz_v,0]
    end

    return Xref, Uref
end
```

Out[5]: `reference_trajectory (generic function with 1 method)`

To solve this problem with Ipopt and `fmincon`, we are going to concatenate all of our x's and u's into one vector (same as HW3Q1):

$$Z = \begin{bmatrix} x_1 \\ u_1 \\ x_2 \\ u_2 \\ \vdots \\ x_{N-1} \\ u_{N-1} \\ x_N \end{bmatrix} \in \mathbb{R}^{N\cdot nx+(N-1)\cdot nu}$$

where $x \in \mathbb{R}^{nx}$ and $u \in \mathbb{R}^{nu}$. Below we will provide useful indexing guide in `create_idx` to help you deal with Z. Remember that the API for `fmincon` (that we used in HW3Q1) is the following:

$$\begin{aligned}
\min_z \quad & \ell(z) && \text{cost function} && (9)\\
\text{st} \quad & c_{eq}(z) = 0 && \text{equality constraint} && (10)\\
& c_L \leq c_{ineq}(z) \leq c_U && \text{inequality constraint} && (11)\\
& z_L \leq z \leq z_U && \text{primal bound constraint} && (12)
\end{aligned}$$

Template code has been given to solve this problem but you should feel free to do whatever is easiest for you, as long as you get the trajectory shown in the animation `walker.gif` and pass tests.

In [19]:
```
# feel free to solve this problem however you like, below is a template for a
# good way to start.

function create_idx(nx,nu,N)
    # create idx for indexing convenience
    # x_i = Z[idx.x[i]]
    # u_i = Z[idx.u[i]]
    # and stacked dynamics constraints of size nx are
    # c[idx.c[i]] = <dynamics constraint at time step i>
    #
    # feel free to use/not use this

    # our Z vector is [x0, u0, x1, u1, …, xN]
    nz = (N-1) * nu + N * nx # length of Z
    x = [(i - 1) * (nx + nu) .+ (1 : nx) for i = 1:N]
    u = [(i - 1) * (nx + nu) .+ ((nx + 1):(nx + nu)) for i = 1:(N - 1)]

    # constraint indexing for the (N-1) dynamics constraints when stacked up
```

```
    c = [(i - 1) * (nx) .+ (1 : nx) for i = 1:(N - 1)]
    nc = (N - 1) * nx # (N-1)*nx

    return (nx=nx,nu=nu,N=N,nz=nz,nc=nc,x= x,u = u,c = c)
end

function walker_cost(params::NamedTuple, Z::Vector)::Real
    # cost function
    idx, N, xg = params.idx, params.N, params.xg
    Q, R, Qf = params.Q, params.R, params.Qf
    Xref,Uref = params.Xref, params.Uref

    # TODO: input walker LQR cost

    J = 0
    for i = 1:(N-1)
        J += 0.5 * (Z[idx.x[i]] - Xref[i])' * Q * (Z[idx.x[i]] - Xref[i])
        J += 0.5 * (Z[idx.u[i]] - Uref[i])' * R * (Z[idx.u[i]] - Uref[i])
    end

    J += 0.5 * (Z[idx.x[N]] - Xref[N])' * Qf * (Z[idx.x[N]] - Xref[N])

    return J
end

function walker_dynamics_constraints(params::NamedTuple, Z::Vector)::Vector
    idx, N, dt = params.idx, params.N, params.dt
    M1, M2 = params.M1, params.M2
    J1, J2 = params.J1, params.J2
    model = params.model

    # create c in a ForwardDiff friendly way (check HW0)
    c = zeros(eltype(Z), idx.nc)

    # TODO: input walker dynamics constraints (constraints 3-6 in the opti problem)
    for i = 1:(N-1)
        xi = Z[idx.x[i]]
        xiplus1 = Z[idx.x[i+1]]
        ui = Z[idx.u[i]]

        if (i in M1) && !(i in J1)
            c[idx.c[i]] .= rk4(model, stance1_dynamics, xi, ui, dt) .- xiplus1
        end
        if (i in M2) && !(i in J2)
            c[idx.c[i]] .= rk4(model, stance2_dynamics, xi, ui, dt) .- xiplus1
        end
        if (i in J1)
            c[idx.c[i]] .= jump2_map(rk4(model, stance1_dynamics, xi, ui, dt)) .- xiplus1
        end
        if (i in J2)
            c[idx.c[i]] .= jump1_map(rk4(model, stance2_dynamics, xi, ui, dt)) .- xiplus1
        end

    end

    return c
end

function walker_stance_constraint(params::NamedTuple, Z::Vector)::Vector
    idx, N, dt = params.idx, params.N, params.dt
    M1, M2 = params.M1, params.M2
    J1, J2 = params.J1, params.J2

    model = params.model

    # create c in a ForwardDiff friendly way (check HW0)
    c = zeros(eltype(Z), N)

    # TODO: add walker stance constraints (constraints 7-8 in the opti problem)
    for i = 1:N
        xi = Z[idx.x[i]]
        if i in M1
            c[i] = xi[4]
        end
        if i in M2
            c[i] = xi[6]
        end

    end

    return c
end

function walker_equality_constraint(params::NamedTuple, Z::Vector)::Vector
    N, idx, xic, xg = params.N, params.idx, params.xic, params.xg

    # TODO: stack up all of our equality constraints
```

```julia
    # should be length 2*nx + (N-1)*nx + N
    # inital condition constraint (nx)       (constraint 1)
    # terminal constraint         (nx)       (constraint 2)
    # dynamics constraints        (N-1)*nx   (constraint 3-6)
    # stance constraint           N          (constraint 7-8)

    return [Z[idx.x[1]] - xic;
            Z[idx.x[N]] - xg;
        walker_dynamics_constraints(params, Z);
        walker_stance_constraint(params, Z)]
end

function walker_inequality_constraint(params::NamedTuple, Z::Vector)::Vector
    idx, N, dt = params.idx, params.N, params.dt
    M1, M2 = params.M1, params.M2

    # create c in a ForwardDiff friendly way (check HW0)
    c = zeros(eltype(Z), 2*N)

    # TODO: add the length constraints shown in constraints (9-10)
    # there are 2*N constraints here

    for i = 1:N
        rb  = Z[idx.x[i]][1:2]   # position of the body
        rf1 = Z[idx.x[i]][3:4]   # position of foot 1
        rf2 = Z[idx.x[i]][5:6]   # position of foot 2

        c[2*i-1] = norm(rb - rf1)
        c[2*i] = norm(rb - rf2)
    end

    return c
end
```

Out[19]: walker_inequality_constraint (generic function with 1 method)

In [20]:
```julia
@testset "walker trajectory optimization" begin

    # dynamics parameters
    model = (g = 9.81, mb= 5.0, mf = 1.0, ℓ_min = 0.5, ℓ_max = 1.5)

    # problem size
    nx = 12
    nu = 3
    tf = 4.4
    dt = 0.1
    t_vec = 0:dt:tf
    N = length(t_vec)

    # initial and goal states
    xic = [-1.5;1;-1.5;0;-1.5;0;0;0;0;0;0;0]
    xg =  [1.5;1;1.5;0;1.5;0;0;0;0;0;0;0]

    # index sets
    M1 = vcat([ (i-1)*10      .+ (1:5)   for i = 1:5]...)
    M2 = vcat([((i-1)*10 + 5) .+ (1:5)   for i = 1:4]...)
    J1 = [5,15,25,35]
    J2 = [10,20,30,40]

    # reference trajectory
    Xref, Uref = reference_trajectory(model, xic, xg, dt, N)

    # LQR cost function (tracking Xref, Uref)
    Q = diagm([1; 10; fill(1.0, 4); 1; 10; fill(1.0, 4)]);
    R = diagm(fill(1e-3,3))
    Qf = 1*Q;

    # create indexing utilities
    idx = create_idx(nx,nu,N)

    # put everything useful in params
    params = (
        model = model,
        nx = nx,
        nu = nu,
        tf = tf,
        dt = dt,
        t_vec = t_vec,
        N = N,
        M1 = M1,
        M2 = M2,
        J1 = J1,
        J2 = J2,
        xic = xic,
        xg = xg,
        idx = idx,
```

```
        Q = Q, R = R, Qf = Qf,
        Xref = Xref,
        Uref = Uref
    )

    # TODO: primal bounds (constraint 11)
    x_l = -Inf*ones(idx.nz) # update this
    x_u =  Inf*ones(idx.nz) # update this
    for i = 1:N
        x_l[idx.x[i]][2] = 0
        x_l[idx.x[i]][4] = 0
        x_l[idx.x[i]][6] = 0
    end

    # TODO: inequality constraint bounds
    c_l = (0.5) * ones(2*N) # update this
    c_u =  (1.5) * ones(2*N) # update this

    # TODO: initialize z0 with the reference Xref, Uref
    z0 = zeros(idx.nz) # update this
    for i = 1:(N-1)
        z0[idx.x[i]] .= Xref[i]
        z0[idx.u[i]] .= Uref[i]
    end
    z0[idx.x[N]] .= Xref[N]

    # adding a little noise to the initial guess is a good idea
    z0 = z0 + (1e-6)*randn(idx.nz)

    diff_type = :auto

    Z = fmincon(walker_cost,walker_equality_constraint,walker_inequality_constraint,
                x_l,x_u,c_l,c_u,z0,params, diff_type;
                tol = 1e-6, c_tol = 1e-6, max_iters = 10_000, verbose = true)

    # pull the X and U solutions out of Z
    X = [Z[idx.x[i]] for i = 1:N]
    U = [Z[idx.u[i]] for i = 1:(N-1)]

    # ------------plotting--------------
    Xm = hcat(X...)
    Um = hcat(U...)

    plot(Xm[1,:],Xm[2,:], label = "body")
    plot!(Xm[3,:],Xm[4,:], label = "leg 1")
    display(plot!(Xm[5,:],Xm[6,:], label = "leg 2",xlabel = "x (m)",
                  ylabel = "y (m)", title = "Body Positions"))

    display(plot(t_vec[1:end-1], Um',xlabel = "time (s)", ylabel = "U",
                 label = ["F1" "F2" "τ"], title = "Controls"))

    # ---------animation-------------
    vis = Visualizer()
    build_walker!(vis, model::NamedTuple)
    anim = mc.Animation(floor(Int,1/dt))
    for k = 1:N
        mc.atframe(anim, k) do
            update_walker_pose!(vis, model::NamedTuple, X[k])
        end
    end
    mc.setanimation!(vis, anim)
    display(render(vis))

    # --------testing----------------

    # initial and terminal states
    @test norm(X[1] - xic,Inf) <= 1e-3
    @test norm(X[end] - xg,Inf) <= 1e-3
#     i = 0
    for x in X
#         i = i+1
#         @show i
        # distance between bodies
        rb = x[1:2]
        rf1 = x[3:4]
        rf2 = x[5:6]
        @test (0.5 - 1e-3) <= norm(rb-rf1) <= (1.5 + 1e-3)
        @test (0.5 - 1e-3) <= norm(rb-rf2) <= (1.5 + 1e-3)

        # no two feet moving at once
        v1 = x[9:10]
        v2 = x[11:12]
        @test min(norm(v1,Inf),norm(v2,Inf)) <= 1e-3

        # check everything above the surface
        @test x[2] >= (0 - 1e-3)
        @test x[4] >= (0 - 1e-3)
```

```
        @test x[6] >= (0 - 1e-3)
    end

end
```

```
---------checking dimensions of everything----------
---------all dimensions good-----------------------
---------diff type set to :auto (ForwardDiff.jl)----
---------testing objective gradient-----------------
---------testing constraint Jacobian----------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve---------------------
This is Ipopt version 3.13.4, running with linear solver mumps.
NOTE: Other linear solvers might be more efficient (see Ipopt documentation).

Number of nonzeros in equality constraint Jacobian...:   401184
Number of nonzeros in inequality constraint Jacobian.:    60480
Number of nonzeros in Lagrangian Hessian.............:        0

Total number of variables............................:      672
                     variables with only lower bounds:        0
                variables with lower and upper bounds:        0
                     variables with only upper bounds:        0
Total number of equality constraints.................:      597
Total number of inequality constraints...............:       90
        inequality constraints with only lower bounds:        0
   inequality constraints with lower and upper bounds:       90
        inequality constraints with only upper bounds:        0

iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
   0  5.8705593e-10 1.47e+00 2.56e-05   0.0 0.00e+00    -  0.00e+00 0.00e+00   0
   1  7.8623691e+01 1.08e+00 3.12e+03  -6.3 1.18e+02    -  2.61e-01 3.55e-01h  1
   2  1.9268562e+02 5.87e-01 3.37e+03   0.4 8.99e+01    -  1.00e+00 4.51e-01h  1
   3  2.1726651e+02 5.18e-01 3.09e+03   0.0 7.01e+01    -  4.66e-01 1.32e-01H  1
   4  2.9064996e+02 3.25e-01 2.29e+03  -0.1 5.98e+01    -  1.00e+00 4.17e-01h  1
   5  4.0595397e+02 4.59e-01 1.77e+02   0.3 3.75e+01    -  4.75e-01 1.00e+00h  1
   6  4.2555082e+02 7.83e-01 1.24e+02  -0.2 6.68e+01    -  4.44e-01 1.00e+00h  1
   7  3.4659206e+02 3.00e-01 1.93e+02   0.4 3.49e+01    -  2.59e-01 1.00e+00f  1
   8  3.0788347e+02 6.52e-02 7.08e+01  -0.0 1.64e+01    -  9.99e-01 1.00e+00f  1
   9  3.1196825e+02 1.04e-01 2.65e+01   0.2 3.33e+01    -  1.00e+00 1.00e+00H  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  10  2.7842587e+02 2.41e-02 2.24e+01  -0.1 1.18e+01    -  9.97e-01 1.00e+00f  1
  11  2.6890116e+02 5.74e-03 4.79e+00  -0.8 1.24e+01    -  9.93e-01 1.00e+00f  1
  12  2.5927308e+02 3.50e-02 6.78e+00  -1.3 3.04e+01    -  9.58e-01 1.00e+00f  1
  13  2.5517338e+02 3.42e-02 4.98e+00  -0.8 8.24e+01    -  1.00e+00 1.69e-01f  3
  14  2.6525675e+02 5.99e-03 6.75e+00  -0.9 2.50e+01    -  9.73e-01 1.00e+00H  1
  15  2.5249420e+02 4.17e-02 6.05e+00  -1.0 9.85e+00    -  9.89e-01 1.00e+00f  1
  16  2.5021834e+02 2.65e-03 1.84e+00  -1.9 5.07e+00    -  9.98e-01 1.00e+00f  1
  17  2.4995351e+02 9.54e-04 1.06e+00  -2.5 1.62e+00    -  1.00e+00 1.00e+00f  1
  18  2.4954295e+02 3.01e-04 9.34e-01  -3.5 1.40e+00    -  1.00e+00 1.00e+00f  1
  19  2.4944842e+02 3.54e-03 2.20e+00  -4.2 4.63e+00    -  1.00e+00 1.00e+00f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  20  2.4879223e+02 4.20e-03 3.15e+00  -4.7 7.84e+00    -  1.00e+00 1.00e+00f  1
  21  2.5008884e+02 1.43e-04 4.30e+00  -5.3 8.27e+00    -  1.00e+00 1.00e+00H  1
  22  2.4823357e+02 1.75e-03 9.75e-01  -5.9 3.65e+00    -  1.00e+00 1.00e+00f  1
  23  2.4853043e+02 1.02e-04 1.26e+00  -6.5 1.93e+00    -  1.00e+00 1.00e+00H  1
  24  2.4789631e+02 1.04e-03 2.00e-01  -7.4 1.83e+00    -  1.00e+00 1.00e+00f  1
  25  2.4788667e+02 5.02e-05 3.05e-01  -8.6 8.76e-01    -  1.00e+00 1.00e+00h  1
  26  2.4783224e+02 2.81e-04 8.47e-01 -10.0 2.04e+00    -  1.00e+00 1.00e+00f  1
  27  2.4781761e+02 4.60e-04 1.36e+00 -10.5 1.52e+01    -  1.00e+00 1.25e-01f  4
  28  2.4781170e+02 6.21e-04 1.41e+00 -11.0 6.31e+00    -  1.00e+00 5.00e-01f  2
  29  2.4778373e+02 4.60e-04 1.29e+00 -11.0 3.18e+00    -  1.00e+00 5.00e-01f  2
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  30  2.4775801e+02 2.04e-04 3.14e-01 -11.0 7.62e-01    -  1.00e+00 1.00e+00f  1
  31  2.4774062e+02 1.97e-05 1.08e-01 -11.0 3.44e-01    -  1.00e+00 1.00e+00h  1
  32  2.4773691e+02 4.86e-06 6.69e-02 -11.0 2.29e-01    -  1.00e+00 1.00e+00h  1
  33  2.4773684e+02 3.65e-05 1.62e-01 -11.0 8.52e-01    -  1.00e+00 1.00e+00h  1
  34  2.4773259e+02 4.56e-05 3.18e-01 -11.0 3.37e+00    -  1.00e+00 2.50e-01h  3
  35  2.4777260e+02 3.30e-07 3.59e-01 -11.0 9.52e-01    -  1.00e+00 1.00e+00H  1
  36  2.4772925e+02 5.72e-05 8.92e-02 -11.0 1.11e+00    -  1.00e+00 1.00e+00f  1
  37  2.4777735e+02 8.95e-05 2.42e-01 -11.0 5.45e-01    -  1.00e+00 1.00e+00h  1
  38  2.4772851e+02 6.78e-05 1.79e-02 -11.0 5.28e-01    -  1.00e+00 1.00e+00f  1
  39  2.4772824e+02 2.73e-08 8.66e-03 -11.0 1.98e-02    -  1.00e+00 1.00e+00h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  40  2.4772792e+02 1.96e-06 7.06e-02 -11.0 1.11e+00    -  1.00e+00 1.25e-01f  4
  41  2.4772870e+02 6.29e-09 6.16e-02 -11.0 2.33e-01    -  1.00e+00 1.00e+00H  1
  42  2.4772814e+02 2.16e-06 4.18e-02 -11.0 9.93e-02    -  1.00e+00 1.00e+00f  1
  43  2.4772753e+02 1.02e-06 2.41e-02 -11.0 1.15e-01    -  1.00e+00 1.00e+00h  1
  44  2.4772769e+02 4.84e-07 1.93e-02 -11.0 2.16e-02    -  1.00e+00 1.00e+00h  1
  45  2.4772752e+02 3.06e-07 2.03e-03 -11.0 2.38e-02    -  1.00e+00 1.00e+00h  1
  46  2.4772752e+02 2.94e-10 1.66e-03 -11.0 1.97e-03    -  1.00e+00 1.00e+00h  1
  47  2.4772755e+02 6.42e-11 1.63e-02 -11.0 4.26e-02    -  1.00e+00 1.00e+00H  1
  48  2.4772753e+02 7.31e-08 2.55e-02 -11.0 4.21e-02    -  1.00e+00 5.00e-01f  2
  49  2.4772752e+02 1.07e-07 6.53e-03 -11.0 4.16e-02    -  1.00e+00 1.00e+00h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  50  2.4772762e+02 2.03e-10 3.42e-02 -11.0 4.97e-02    -  1.00e+00 1.00e+00H  1
  51  2.4772751e+02 2.36e-07 2.45e-03 -11.0 5.60e-02    -  1.00e+00 1.00e+00f  1
  52  2.4772756e+02 7.40e-08 1.75e-02 -11.0 2.50e-02    -  1.00e+00 1.00e+00h  1
  53  2.4772751e+02 6.23e-08 1.15e-03 -11.0 2.25e-02    -  1.00e+00 1.00e+00h  1
  54  2.4772751e+02 4.08e-10 2.11e-03 -11.0 2.68e-03    -  1.00e+00 1.00e+00h  1
  55  2.4772751e+02 2.29e-10 3.41e-04 -11.0 1.59e-03    -  1.00e+00 1.00e+00h  1
  56  2.4772751e+02 5.23e-11 1.76e-04 -11.0 3.42e-04    -  1.00e+00 1.00e+00h  1
```

```
  57  2.4772751e+02 2.74e-11 4.09e-04 -11.0 7.88e-04    -  1.00e+00 1.00e+00h  1
  58  2.4772751e+02 1.49e-12 3.14e-03 -11.0 9.46e-03    -  1.00e+00 1.00e+00H  1
  59  2.4772751e+02 1.04e-08 1.81e-04 -11.0 5.93e-03    -  1.00e+00 1.00e+00f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  60  2.4772751e+02 1.07e-11 1.06e-04 -11.0 3.16e-04    -  1.00e+00 1.00e+00h  1
  61  2.4772751e+02 7.84e-16 1.88e-04 -11.0 3.50e-04    -  1.00e+00 1.00e+00H  1
  62  2.4772751e+02 8.45e-12 4.37e-05 -11.0 2.23e-04    -  1.00e+00 1.00e+00h  1
  63  2.4772751e+02 8.89e-13 3.61e-05 -11.0 1.37e-04    -  1.00e+00 1.00e+00h  1
  64  2.4772751e+02 2.66e-15 6.69e-04 -11.0 2.25e-03    -  1.00e+00 1.00e+00H  1
  65  2.4772751e+02 1.87e-10 1.22e-04 -11.0 1.26e-03    -  1.00e+00 1.00e+00h  1
  66  2.4772751e+02 7.62e-11 3.67e-04 -11.0 7.35e-04    -  1.00e+00 1.00e+00h  1
  67  2.4772751e+02 5.86e-11 3.84e-05 -11.0 5.98e-04    -  1.00e+00 1.00e+00h  1
  68  2.4772751e+02 9.75e-13 3.05e-05 -11.0 5.75e-05    -  1.00e+00 1.00e+00h  1
  69  2.4772751e+02 1.73e-13 6.59e-06 -11.0 2.35e-05    -  1.00e+00 1.00e+00h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  70  2.4772751e+02 8.48e-14 8.71e-06 -11.0 1.74e-05    -  1.00e+00 1.00e+00h  1
  71  2.4772751e+02 4.55e-13 2.11e-05 -11.0 8.39e-05    -  1.00e+00 1.00e+00h  1
  72  2.4772751e+02 6.66e-16 6.52e-05 -11.0 4.82e-04    -  1.00e+00 1.00e+00H  1
  73  2.4772751e+02 4.72e-16 1.06e-04 -11.0 2.65e-04    -  1.00e+00 1.00e+00H  1
  74  2.4772751e+02 1.11e-11 3.39e-05 -11.0 4.55e-04    -  1.00e+00 1.00e+00h  1
  75  2.4772751e+02 5.55e-12 2.58e-05 -11.0 1.25e-04    -  1.00e+00 5.00e-01h  2
  76  2.4772751e+02 4.33e-13 1.06e-05 -11.0 9.99e-05    -  1.00e+00 1.00e+00h  1
  77  2.4772751e+02 2.64e-13 1.51e-05 -11.0 3.58e-05    -  1.00e+00 1.00e+00h  1
  78  2.4772751e+02 1.49e-13 1.86e-06 -11.0 3.20e-05    -  1.00e+00 1.00e+00h  1
  79  2.4772751e+02 2.31e-14 3.43e-06 -11.0 5.56e-06    -  1.00e+00 1.00e+00h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  80  2.4772751e+02 1.09e-14 3.37e-07 -11.0 3.58e-06    -  1.00e+00 1.00e+00h  1

Number of Iterations....: 80

                                   (scaled)                 (unscaled)
Objective...............:   2.4772750671476896e+02    2.4772750671476896e+02
Dual infeasibility......:   3.3732523109508694e-07    3.3732523109508694e-07
Constraint violation....:   1.0880185641326534e-14    1.0880185641326534e-14
Complementarity.........:   1.0000000000000003e-11    1.0000000000000003e-11
Overall NLP error.......:   3.3732523109508694e-07    3.3732523109508694e-07


Number of objective function evaluations             = 138
Number of objective gradient evaluations             = 81
Number of equality constraint evaluations            = 138
Number of inequality constraint evaluations          = 138
Number of equality constraint Jacobian evaluations   = 81
Number of inequality constraint Jacobian evaluations = 81
Number of Lagrangian Hessian evaluations             = 0
Total CPU secs in IPOPT (w/o function evaluations)   =    105.402
Total CPU secs in NLP function evaluations           =     15.324

EXIT: Optimal Solution Found.
```





┌ **Info:** MeshCat server started. You can open the visualizer by visiting the following URL in your browser:
└ http://127.0.0.1:8709

Open Controls



```
Test Summary:                   | Pass  Total
walker trajectory optimization |  272    272
```

Out[20]: `Test.DefaultTestSet("walker trajectory optimization", Any[], 272, false, false)`

In [ ]:

In [ ]: